'##################################################
'###  Top Level (Get root Path from system env)
'##################################################
function GetEnv(envVarName)
	On Error Resume Next
	' 创建Shell对象并获取值
	Set WshShell = CreateObject("WScript.Shell")
	envValue = WshShell.Environment("PROCESS")(envVarName) ' 优先查找进程级变量

	' 如果未找到则尝试系统级变量
	If Err.Number <> 0 Or envValue = "" Then
		Err.Clear
		envValue = WshShell.Environment("SYSTEM")(envVarName)
	End If

	' 如果仍未找到则尝试用户级变量
	If Err.Number <> 0 Or envValue = "" Then
		Err.Clear
		envValue = WshShell.Environment("USER")(envVarName)
	End If

	' 检查结果并输出
	If Err.Number = 0 And envValue <> "" Then
		GetEnv=envValue
	Else
		GetEnv=""
	End If
end function

'##################################################
'##################################################
'##################################################
sub IncludeVbs(filepath)
	Dim fso, file, content
	Set fso = CreateObject("Scripting.FileSystemObject")
	' 读取functions.vbs的内容
	Set file = fso.OpenTextFile(filepath, 1) ' 1表示只读
	content = file.ReadAll
	file.Close
	' 将functions.vbs的内容载入全局作用域
	ExecuteGlobal content
end sub

'##################################################
'##################################################
'##################################################
dim root,tmpfile
Dim doc
Set doc = application.ActiveDocument
dim dbDict:set dbDict = createobject("scripting.Dictionary")
dim fso:set fso = createobject("scripting.filesystemobject")	
dim color_def
main
set doc = nothing
set dbDict=nothing
set fso = nothing

sub main()	
	'// REUSE
	envVarName ="EXP_SHORTCUTS_ROOT"
	root = GetEnv(envVarName)'"C:\MentorGraphics\2005EXP\SDD_HOME\pvtk\env\"
	if root ="" then 
		msgbox "please set env value (EXP_SHORTCUTS_ROOT = ?)"
		exit sub
	end if
	
	call IncludeVbs(root & "\skill_lib.vbs")
	
	If ValidateServer(doc) = 0 Then
	    msgbox "Server validation failed."
	    exit sub
	End If	
	
	dim r,g,b,index
	r = 255
	b = 255
	g = 255
	index = 1
	dim colorPatternObj
	if application.version = "EXP2005.1_060615.00 2005.1" then
		set colorPatternObj = application.Utility.NewColorPattern(r, g, b,index)
	else	
		set colorPatternObj = application.Utility.NewColorPattern(r, g, b,100,index)
	end if
	
	dim pcbnets,pcbnet
	set pcbnets = doc.nets(1)
	For each pcbNet In pcbNets
		If pcbNet.Selected = True Then
			pcbNetName = pcbNet.Name
			doc.activeView.DisplayControl.Global.Net("True",pcbNet)=false
			set doc.activeView.DisplayControl.Global.NetColor("True",pcbNet) = colorPatternObj				
		end if
	next	
end sub
